
Why declarative problem-solving?

e Solving problems with a computer

Problem Solution

Representing Interpreting

Computer
Generating

Why declarative problem-solving?

e Solving problems with a computer by programming

Problem Solution

Procedural programming Interpreting

Program
Executing

« A procedural program specifies “how to generate solutions”
« Solutions are generated by executing a program

Why declarative problem-solving?

e Solving problems with a computer by a declarative specification

Problem Solution

Modeling Interpreting

Specification
Solving

« A declarative program specifies “what are the solutions”
« Solutions are generated by a problem solver

Why declarative problem-solving?

e E.g., configuration of technical system aka automated engineering

© Siemens

Why declarative problem-solving?

e Some insights of applying declarative problem-solving:

Total cost for configurator kernel Mio €

Mio € employing procedural programming

Years

Initial

Maintenance cost: approx. 15% per year of initial development cost

Why declarative problem-solving?

Total cost for configurator kernel Mio €
employing declarative programming

Mio € 14 * Reduction of initial development cost by 66%
12 * Reduction of yearly maintenance cost by 80%
1g ° Productivity increase by 300% (no additional staff)
6 * ROl in1year for the telecommunication domain
4 * Enhanced user interaction:
2  explanations,
0 Nt ©O®OoONTO® o ¥ Years * incremental configuration,

Initial

* repair...

Wanted

e General-purpose framework for modeling and solving problems

Design requirements

e “Expressive” to succinctly formulate maintainable specifications
e Optimization

e Rules and constraints

e Aggregation

e Dealing with the absence of information

Proposal
e Answer set programming (ASP) paradigm

www.aau.at 'll

Preliminaries |/ brush-ups

Propositional logic (syntax):

e Finite set of atoms (propositional symbols): e.g., {a,, a,, b, ..., name_of(4711,joe), age of(4711,20), ...}
e Logic operators: “V”, “N\”, ¢ 470

e Literals are atoms and negated atoms: e.g., a, -a

o Aclauseis aset of literals connected by “v”’: e.g., (-b v a, vV ¢)

e Propositional theory: a finite set of clauses connected by “A”: e.g., (= 2a,Va,) AbA(-bVva,Vc)

www.aau.at 'l'

Propositional logic (semantics)

e Letthe base B of a propositional theory T be the set of atoms of T
e Aninterpretation | is a subset of a base B, i.e., | € B

e There are two truth values, i.e., true and false. An interpretation | associates either true or false
to all atoms of B, i.e., all atoms in | are true, all atoms in B \ | are false

e Thelogic operators (“V”, “A”, “=", “=””) are functions mapping truth values of their arguments to
a truth value: e.g., (true A false) - false, = true - false. Operators have the usual definition

e Given aninterpretation | the truth value of a proposition theory T can be determined by
recursively evaluating the logical operators, i.e., eval: { | } x{ T } = {true, false}

e Aninterpretation | is a model of a propositional theory T iff eval(l,T) = true

www.aau.at "l

Example

e Propositional theory
T:(—|a1va2)/\b/\("bVa1VC)

e ThebaseofT
B={a1)a27bic}

e |Interpretation {a,, b }is not a model of T, because (- a,V a,) is evaluated to false

* Interpretations {a,, a,,b }and {b, c} are models of T, because all clauses of T are evaluated to true

www.aau.at 'll

Propositional logic programs

“a < b”is equivalentto “aVv -b”, hence clausec: a,v...va,v-b,v..v-b, isequivalentto

rule r: (a,vV..VvVa,) < (b,A...Ab,,) writtenin ASP as
a,|...|ag:-b,, .., b,

In answer set programming

A propositional theory is called a (logic) program

Clauses are rules . Rules are ended by “.”

“a,v...va/ is called the head of rule r
“b, A... Ab,,” is called the body of rule r

“»

“A” corresponds to ,

l”

“v”’ corresponds to “
“<" corresponds to “:-”
Arule “a;.” with exactly one atom in the head and no atoms in the body is called a fact

Arule “« b, A... Ab,.” with empty head is called a constraint

www.aau.at 'll

Rules in ASP

Logic program P comprises a set of rules:

a,|a,| - -|ax:=-b,...,b, notc,...,notc,.

e.g., head(X) | tail(X) :- coin(X), flippedCoin(X), not edge (X).
* 3, b;qareatoms
e “not” is called negation as failure (naf) or default-negation

e All-quantified logical variables are allowed, but first we focus on the propositional case

e Rules must be “safe” (required for transforming all-quantified rules to propositional logic)
- Roughly speaking, all variables in a;, ¢;must be contained in some b;(more details follow)

Rules in ASP

e.g., p(a,X) :- g(¥,1), k("a_string"), X =Y + 20, X > 30, not exception.

Terms are constants, variables, arithmetic terms. For simplicity no functional terms

Constants are

— symbolic constants (strings starting with some lowercase letter)
— string constants (quoted strings)
- integers

e Variables are strings starting with some uppercase letter. Variables are all-quantified
e Arithmetic terms have the form —(t) or (t ¢ u) for terms t and u with o € {47 =7 “x” « [’}

e C(lassical atoms have the form p(t,,...,t;) where t;is a term and p is a predicate name, staring with
some lowercase letter
— p() with arity o is a classical atom. Parentheses can be dropped

e Built-in atoms have the form t < u for terms t and u with < € {“<”, “<”, =7 g7 57«71

) ) ) )

e Built-in atoms a as well as the expressions a and not a for a classical atom a are naf-literals
o Aggregate atoms will be defined later

www.aau.at "l

Dinner example

Wine bottles (brands) 3, ... ,e

e Plain ontology natively represented within the logic program
e Preference by facts

% A suite of wine bottles and their kinds

wineBottle (a) . isA(a,whiteWine). isA (a,sweetWine) .
wineBottle (b) . isA (b,whiteWine) . isA (b,dryWine) .
wineBottle (c) . isA(c,whiteWine). isA(c,dryWine) .
wineBottle (d) . isA(d, redWine) . isA(d,dryWine) .
wineBottle (e) . isA (e, redWine) . isA (e, sweetWine) .

% Persons and their preferences

person (axel) . preferredWine (axel,whiteWine) .
person (gibbi) . preferredWine (gibbi, redWine) .
person (roman) . preferredWine (roman,dryWine) .

% Available bottles a person likes
compliantBottle (X,Z) :- preferredWine (X,Y), isA(Z,Y).

www.aau.at "l

Default negation # classic negation

“not a” is negation as failure (default negation)
e We assume assertion not a as true, if there is no reason to “believe” in a
e There are no unnecessary facts in the model. l.e., a is not a fact or cannot be deduced

Example:
compliantBottle (axel,a).
bottleChosen(a) :- not bottleSkipped(a),

compliantBottle (axel, a).

bottleChosen(a) | bottleSkipped{a)r——compriantBottle (axel, a).

Preferred minimal model:

M1 = compliantBottle (axel,a), bottleChosen(a) }

- eSkipped(a) }.

www.aau.at 'll

Programs with negation

Extension of example:

compliantBottle (axel,a).

bottleChosen (X) :— not bottleSkipped(X), compliantBottle (Y, X).
bottleSkipped (X) :— not bottleChosen(X), compliantBottle(Y,X).
Result ?77?

Problem: no single minimal model

Two alternatives:
e M1= { compliantBottle (axel,a), bottleChosen(a) },

e M2= { compliantBottle (axel,a), bottleSkipped(a) }.

Which one to choose?

www.aau.at "l



'llUNIVERSITAT
KLAGENFURT

Social dinner example cont.

Extend the simple social dinner example:

% These rules generate multiple answer sets:

(1) bottleSkipped(X) :- not bottleChosen (X),
compliantBottle (Y, X).

(2) bottleChosen (X) :- not bottleSkipped (X),
compliantBottle (Y,X).

(3) hasBottleChosen(X) :- bottleChosen(Z), compliantBottle(X,Z).

e Rules (1) and (2) enforce that either bottleChosen (X) or bottleSkipped (X) isincludedin an
answer set (but not both), if it contains compliantBottle (Y, X)

e Rule (3) computes which persons have a bottle

www.aau.at 'll

Social dinner example cont.

% Alternatively, we could use disjunction:
(4) bottleSkipped(X) | bottleChosen(X) :- compliantBottle(Y,X).

(3) hasBottleChosen (X) :- bottleChosen(Z), compliantBottle (X, Z).

e Rules (1) and (2) enforce that either bottleChosen (X) Or bottleSkipped (X) is included in an
answer set (but not both), if it contains compliantBottle (¥, X)

e Rule (3) computes which persons have a bottle
e Rule (4) (disjunction!) can be used for replacing (1)-(2), more in the appendix

www.aau.at 'll

Answer Set Semantics

e Variable-free, non-disjunctive programs first!
e Normal Rules
a:-b,...,b,,notc,...,notc,
where all 3, b;, ¢; are atoms
e Anormallogic program P is a (finite) set of such rules

e HB(P) (Herbrand Base) is the set of all atoms with predicates and constants from P

www.aau.at 'll

Example

compliantBottle (axel,a).

wineBottle (a) .

bottleSkipped(a) :— not bottleChosen(a), compliantBottle (axel,a).
bottleChosen (a) :— not bottleSkipped(a), compliantBottle (axel,a).
hasBottleChosen (axel) :—- bottleChosen(a), compliantBottle (axel,Ka).
e HB(P)={

wineBottle (a) , wineBottle (axel),
bottleSkipped(a) , bottleSkipped(axel),
bottleChosen(a) , bottleChosen (axel),
hasBottleChosen(a) , hasBottleChosen (axel),
compliantBottle (axel,a), compliantBottle (axel,axel),
compliantBottle(a,a), compliantBottle(a,axel) }

www.aau.at 'll

Answer sets

Let

e P beanormallogic program
e M S HB(P) be a set of atoms

Gelfond-Lifschitz (GL) Reduct PM

The reduct PM is obtained as follows (based on “guessed” M):

©® remove from P each rule

a:-b,...,b,, notc,...,notc, wheresome ¢ isin M

® remove all literals of form not ¢; from all remaining rules

www.aau.at 'll

Answer sets

e Thereduct PMis a Horn program (clauses with at most one positive literal, i.e., facts and rules where the
head may be empty)

e It has the least model Im(PM). There exists at most one minimal model

Definition:
M € HB(P) is an answer set of P if and only if M = Im(P™)

Intuition:

e M makes an assumption about what is true and what is false
o PMderives positive facts under the assumption for which atom not (-) is true as defined by M
e Iftheresultis M, then the assumption of M is “stable”

www.aau.at 'll

Computation of Im(P)

The least model of a not-free program can be computed by fixpoint iteration

Algorithm Compute LM(P)
Input: Horn program P;
Output: Im(P)

new M :=J;
repeat
M = new_M,;

new M:={a|“a:-b,,...,b,” €P,{b,...,b, .} E M}
until new M ==M
return M

www.aau.at 'll

Examples 1

compliantBottle (axel,a).
wineBottle(a) .
hasBottleChosen (axel) :- bottleChosen(a), compliantBottle (axel,a).

e Phasnonot(i.e., is Horn)
e thus, PM =P forevery M
e thesingle answer set of Pis
M =Im(P)= { wineBottle(a), compliantBottle (axel,a) }.

www.aau.at 'll

Examples 2

(1) compliantBottle (axel,a).
wineBottle(a).

(2) bottleSkipped(a) - compliantBottle (axel,a).
(3) not bottleSkipped(a),
(4) hasBottleChosen (axel) :— bottleChosen(a), compliantBottle (axel,a).

Take M = { wineBottle(a), compliantBottle (axel,a), bottleSkipped(a) }

e Rule (2) “survives” the reduction (delete not bottleChosen(a))
e Rule (3)is dropped (not bottlesSkipped(a) is false)

Im(PM) = M, and thus M is an answer set

www.aau.at 'll

Examples 3

(1) compliantBottle (axel,a).
wineBottle (a).

(2) not bottleChosen(a), .
(3) bottleChosen(a) - compliantBottle (axel,Ka).
(4) hasBottleChosen (axel) :— bottleChosen(a), compliantBottle (axel,a).

Take[W=={ wineBottle (a) , compliantBottle (axel,a), bottleChosen(a), hasBottleChosen (axel) }

e Rule (2)is dropped
e Rule (3) “survives” the reduction (delete not bottleSkipped(a))

Im(PM) = M, and therefore M is another answer set

www.aau.at 'll

Examples 4

(1) compliantBottle (axel,a).
wineBottle (a).

(2) not bottleChosen (a)
(3) not bottleSkipped(a)
(4) hasBottleChosen (axel) :— bottleChosen(a), compliantBottle (axel,a).

Take M = { wineBottle(a), compliantBottle (axel,a),
bottleChosen(a) , bottleSkipped(a), hasBottleChosen (axel) }

e Rules(2)and (3) are dropped

hn(PM)=={ wineBottle(a), compliantBottle(axel,a)} M
Thus, M is not an answer set

www.aau.at 'll

Examples 5

(1) compliantBottle (axel,a).
wineBottle(a).

(2) bottleSkipped(a) - compliantBottle (axel,a).
(3) bottleChosen (a) - compliantBottle (axel,Ka).
(4) hasBottleChosen (axel) :— bottleChosen(a), compliantBottle (axel,a).

Take M = { wineBottle(a), compliantBottle (axel,a)}

e Rule (2) “survives” the reduction (delete not bottleChosen(a))
e Rule (3) “survives” the reduction (delete not bottleSkipped(a))

hn(PM)::{ wineBottle (a) , compliantBottle (axel,a), bottleSkipped(a), bottleChosen(a),
hasBottleChosen (axel) } # M

Thus, M is not an answer set

www.aau.at "l

Programs with variables

e Eachruleis ashorthand for all its ground substitutions, i.e., replacements of variables with
ground terms (variable-free, e.g., constants)
e Assumption: number of answer sets and the size of models are finite

e Assured, e.g., by syntactic restrictions such as no function symbols
e For simplicity we limit ground terms to constants

E.g., “b(X) :- not s(X), c(¥,X).” with constants axel and a is a shorthand for:

b(a) :- not s(a), c(a,a).
b(a) :- not s(a), c(axel,a).

b(axel) :- not s(axel), c(axel,axel).
b(axel) :- not s(axel), c(a,axel).

www.aau.at "l

Programs with variables

e The Herbrand base of program P, HB(P), consists of all ground (variable-free) atoms with predicates and
constant symbols from P

e The grounding of a rule r, Ground(r), consists of all rules obtained from r if each variable in r is replaced by
some ground term (over P, unless specified otherwise)

* The grounding of program P, is Ground(P) = U, Ground(r )

Definition:
M € HB(P)is an answer set of P if and only if M is an answer set of Ground(P)

www.aau.at 'll

Inconsistent programs

Program
P :- not p.

e This program has NO answer sets, both guesses { p } and { } arenotanswer sets

e LetPbeaprogramand p be a new atom
e Adding

P :- not p.

to program P “kills” all answer sets of P

www.aau.at 'll

Constraints

Adding
P :- 9,---y 9n , Dot r;, ..., not r,, not p.

to P “kills”” all answer sets of P that:
e containg;,..., g, and
e donotcontainr,, ..., r,
e Abbreviation:

- d1,--+7 9n , Dot r;,..., not r,.

e Thisis called a “constraint” (cf. integrity constraints in databases)
e A constraintisviolated,ifq,, ..., g, , not r;,..., not r_issatisfied

www.aau.at 'l'

Social dinner example (cont.)

Task
e Add a constraint in order to filter answer sets in which for some person no bottle is chosen

% This rule generates multiple answer sets:
(1) bottleSkipped(X) :- not bottleChosen(X), compliantBottle(Y,X).
(2) bottleChosen (X) :— not bottleSkipped(X), compliantBottle(Y,X).

% Ensure that each person gets a bottle.
(3) hasBottleChosen (X) :- bottleChosen(Z), compliantBottle(X,Z).
(4) :- person(X), not hasBottleChosen (X) .

www.aau.at 'll

Main reasoning tasks

Consistency: decide whether a given program P has an answer set

Cautious (resp. brave) reasoning: given a program P and ground literals |, ..., |,
decide whether |, ..., |, simultaneously hold in every (resp., some) answer set of P

Query answering: Given a program P and non-ground atom a on variables X, .. ., X,
list all assignments of values v to X,, . . ., X, such that substituting the variables in a with v is

cautiously resp. bravely true

e Seamless integration of query language and rule language

Answer Set Computation: compute some [ all answer sets of a given program P

www.aau.at "l

Properties of Answer Sets

Minimality:
Each answer set M of P is a minimal model (w.r.t €)

Generalization of stratified semantics:

If negation in a normal logic program P (no disjunction) is layered (‘P is stratified”),
then P has a unique answer set, which coincides with the perfect model

NP-Completeness:

Deciding whether a normal propositional program P has an answer set is NP-complete in general
= Answer Set Semantics is an expressive formalism

e Higher expressiveness through further language constructs (disjunction, weak/weight constraints)
e Computational complexity is beyond NP

www.aau.at 'll

Answer set programming paradigm

General idea: Models are solutions

Reduce solving a problem instance | to computing models

Problem

Instance

.| Encoding:

Program P

Theory

>

ASP
Solver

Model(s)

Solution(s)

@ Encode | as alogic program P, such that solutions of | are represented by models of P

® Compute some model M of P, using an ASP solver

© Extract asolution for | from M

Variant: Compute multiple models (for multiple resp. all solutions)

www.aau.at 'll

ASP in practice

Problem . Encoding:

spec. (PS) | Program Pes | 1y oo [ AP Model(s)
oo [ Encoding Solver Solution(s)
(D) Program P

Uniform encoding:
Separate problem specification PS and input data D (usually, facts)

e Compact, easily maintainable representation: logic programs with constraints
e Integration of knowledge representation, databases, and search techniques

www.aau.at 'll

Architecture of ASP Solvers

Typically, a two-level architecture

1. Grounding Step

e Input: a program P with variables

* Generate a (subset of its) grounding which has the same models

e OQOutput: a program P with no variables

The safety property of rules is exploited by grounding to compress the resulting propositional ASP program

www.aau.at 'll

Architecture of ASP Solvers /2

2. Model search
This is applied for ground programs

Techniques:
e Translations to SAT (propositional satisfiability)

e Special search procedures for ASP such as

— Conflict-driven answer set solving: From theory to practice, M. Gebser, B. Kaufmann, T. Schaub.
Al Journal 187-188 (2012), Elsevier

— Backtracking procedures for assigning truth value to atoms
— Similar to DPLL algorithm for SAT solving

e Important: Heuristics (which atom/rule to assign a truth value)

Exception: Lazy grounding (e.g., Alpha system) to save memory consumption

www.aau.at "l

Answer Set Solvers

e (lingo https://[potassco.org/clingo/

e Cmodels https://www.cs.utexas.edu/users/tag/cmodels/
e DLV https://dlv.demacs.unical.it/home

* Smodels http://www.tcs.hut.fi/Software/smodels/

o Alpha (lazy grounding) https://github.com/alpha-asp/Alpha#

(see https://en.wikipedia.org/wiki/Answer_set_programming for a more extensive list)

Applications of ASP

e configuration

e scheduling

e routing

e diagnosis

e security analysis

e computer-aided verification

Extensions of ASP

e Many extensions of normal logic programs have been proposed such as
— Weak constraints
— Aggregation
— Disjunction

e Some of these extensions are motivated by applications

e Some of these extensions are syntactic sugar, other strictly add expressiveness

www.aau.at 'll

Weak constraints

~ Qyyeesy m , NOL 1, ...y NOt 1, [Weight@Level, t,,...,t,]

Syntax:

e Weight and Level are integers or variables bound in q,,..., g,

e t,.,t areterms, e.g., constants or variables bound in q,,..., q,,

e (@WLevel may be omitted. In this case Level = 0. t,,...,t, may be omitted

Semantics:

e Let PRIOS be the set of all Weight(@Level, t,,...,t, elements of weak constraints in a ground
program P and an answer set A of P, where q,,..., q,,, notr,, ..., not r, is satisfied in A

e Note, all elements in PRIOS are ground. PRIOS is a set. Duplicates are removed

e The answer sets of program P are those answer sets of P which
minimize the sum of the weights of the violated weak constraints

e Minimize the violation of high-level constraints first
e Weak constraints provide means for optimizing objective functions

Weak constraints — Examples

a :- not b. b :- not a. c :- b.
:~ a. [1@0,1]
:~ b. [1@0,1]
:~ c. [1@0,1]

Optimal answer sets: {a}, {b,c} with sum of weights 1atlevel o

a :- not b. b :- not a. c :- b. d :- b.
i~ a. [3,a]
:~ b. [1,b]
i~ c. [1,cd]
i~ d. [1,cd]

Optimal answer set: {b,c,d} with sum of weights 2 at level o

www.aau.at 'll

Aggregate atoms

Aggregation allows to express properties over a set of literals that are true in a model

Example: Given a set of facts defining prices of wines, find the price of the most expensive one

costs(a,l). costs(b,7). costs(c,5). costs(d,2). costs(e,b).
expensiveWine (Y) :- Y = #max{C : costs(W,C)}.

[o]

% grounded and evaluated to

expensiveWine (7) :- 7 = #max{l:costs(a,l); 7:costs(b,7); 5:costs(c,5);
2:costs(d,2); 5:costs(e,b)}.

« Y = #max{C : costs(W,C)} isanaggregate atom

- #max is a build-in aggregate function defined over
- {C : costs(W,C)} acollection of aggregate elements
- C : costs(W,C) isan aggregate element

Aggregate atoms

Aggregate atoms: u,<, #aggr E <, u, (either u,<, or <, u, may be omitted)

e u,and u,are terms (i.e., variables or integers)
Aggregate relation: < E {“<”,“S”’“=”,“¢”,“>”,“2”}
Aggregate function: #aggr € {“#count”, “#sum”’, “#max”’, “#min’’}

e Eisacollection of aggregate elements separated by *;”
Aggregate element: ooy tiniliyeess |y

° t,..t,areterms (e.g., variables or constants)
e |,..,l,are naf-literals (atom a; or not a;)
e Aggregate elements are instantiated during grounding

Satisfaction of aggregate atoms

E.g., aggregate atom: #sum{C,W : costs(W,C)} =Y
Aggregate element (E): C,W : costs(W,C)

Facts: costs(c,5). costs(e,b).

Interpretation | ={costs(c,5), costs(e,5)}
Instantiation of E: E={5,c:costs(c,5); 5,e:costs(e,5)}

— Given a collection E of aggregate elements and an interpretation | € HB(P) of program P
— eval(g,1) = {(t,-..,tm) | tyeeetin = Iyeessl, OCcurs in E and 1,,..., 1, are true with respect to I}

E.g., evaluation of E w.r.t. I: eval(E,)={(5,c), (5,e)}

— Aggregate atom #aggr E < u is true (or false) with respect to | if #aggr(eval(E,l)) < uis true
(or false) with respect to |

— #aggris applied on the first elements of the the tuples in eval(E,l)

E.g., application of aggregate function:  #sum({(5,¢c), (5,e)})=10

Satisfaction of aggregate atoms

Notes:

e Truth of u < #aggr E is analogy defined

* The aggregate function #aggr is applied on the set provided by eval(E,)
e Duplicates are removed!

Examples

costs(a,l). costs(b,7). costs(c,5). costs(d,2). costs(e,b).
expensiveWine (Y) : - #max{C : costs(W,C)} = Y.
Instantiation of aggregate element “c : costs(w,c)”:

° {1:costs(a,1); 7:costs(b,7); 5:costs(c,5); 2:costs(d,2); 5:costs(e,5)}

Evaluation of instantiated aggregate element is
o {m, (M, (5), (2), 453}

Application of #max on evaluation result provides 7

- expensiveWine (7)is true

Examples of aggregate atoms

q :- 0 <= #count{X,Y : a(X,2,k),b(1,2,Y)} <= 3.
qg(2) :- 2 < #sum{V : d(V,2)}, c(2).
P(W) :- #min{S : c(S); T: d(T)} = W.

= #max{V : d(V,2)} > G, c(G).

e leta,|a,|---|ac:-b,...,b,bearulerwhereb,...,b,arenaf-literals

e Avariableis global in a rule or weak constraint r, if it appears outside of aggregate elementsinr

e ForasetV of variables and literals b,,...,b,, v € Vis bound by b,,...,b,
if v occurs outside of arithmetic terms in some b, for 1<i<n such that b, is
— aclassical atom, i.e., b; is not negated, no built-in atom and no aggregate atom, or

— abuilt-in atom t=v or v=t, and any member of V occurring in t is bound by {b,,...,b, }\b;
(e.g., t can be an arithmetic term, a constant or a variable) or

— anaggregate atom #aggr E = v, and any member of V occurring in E is bound by {b.,...,b, }\b;
e The entire setV of variables is bound by b,,...,b if each v € Vis bound by b,,...,b,

e Arule or weak constraint r is safe if the set V of global variables in r is bound by b,,...,b,, and for each
aggregate element t,,...,t, : I,,...,I, in r with occurring variable set W, the set W \V of local variables is
bound by |,,...,I,

Examples

Safe rules and constraints

e a(X) :- node(X), #count{V : edge(V,X)} > 0.

e a(X) :- node(X), not #count{V : edge(V,X)} = 0.

e a(X) :- #fcount{V : node(V), succ(V,Z), not node(Z)} = X.
e :- #count{V : edge(V,Y), not edge(Y,V)} =X, X > 2.

e :- not node(X), #count{V : edge(V,Y)} = X.

Unsafe rules and constraints

e a(X) :- not node(X), #count{V : edge(V,X)} > 0.

e a(X) :- node(X), #count{V : edge(V,X)} > Z.

e a(X) :- node(X), #count{V : edge(V,X), not edge(V,Y)} > 0.
e a(X) :- #count{V : node(V), not edge(V,Y), ¥Y=V+Z} > 0.

e :- #count{V : edge(V,Y), not edge(Y,X)} > 0, X > 2.

e :- #count{V : edge(V,Y)} > 0, X > Y.

e :- not node(X), #count{V : edge(V,Y)} > X.

Clingo

(one of the) most efficient ground-and-solve
ASP systems

Various special language constructs

Conditional Literals:

Givenr(a), r(b), r(c) asfacts
e P :- gq(X) : r(X). standsfor
* p - g(a), g(b), gq(c).

Choice rules:

* 2 {p(X,Y) : q(X)} 7 :- r(Y).

Conditional literals

Form of conditional literals:

A, A, ..., A, where A is a literal and may contain variables

e Instantiate the “head literal” A, where the instantiations of the condition A, ..., A, is true

e The predicates of literals on the right-hand side of a colon (:) are usually defined from facts
without any negative recursion, i.e., these facts can be fully evaluated by the grounder

e A conditional literal is terminated by “;” when further literals in the rule body follow

Conditional literals

Example

person(jane) . important(jan). available(jane).
person(john) . important(john). available(john).

person (sam) .

schedule.

meet :- available(X) : person(X), important (X) ;schedule.
% rule for meet is grounded to

meet :- available(jane), available(john), schedule.

Choice rules

By choice rules, we choose a subset of classical atoms of the head:
A A AL, - AL NOtA L, ..., NOtA,.

e A ;...;A,areclassical atoms or conditional literals where the head is a classical atom

e If body is satisfied in an answer set
then any subset of {A, ;... ; A} can be included in the answer set

e This behavior can be implemented by generating new symbols and rules

Example: {a} :- b. b.

has two answer sets, i.e.,
{b} and
{a,b}

Choice rules with restrictions

We can add optionally lower and upper bounds or restrictions to choice rules:
W< {A ;A< U AL, - A DOt A, L, ..., NOt A,

o Aggregate relation: < € {<”,“<” (=7 “g? (57 4571 Dif omitted then “<” is applied

e u,and u,are terms (i.e., variables or integers)

If body is satisfied in an answer set, then any subset of {A, ;... ; A,} can be included in the answer
set but the restrictions on the cardinality of the subset must be satisfied

This behavior can be implemented by generating new symbols, rules, and aggregate constraints

Choice rules

Example

c(2).
d(a) .d(b) .

1{a(Y):d(Y);b(2):d(2)} < X:- c(X).

Answer: 1
b (b)
Answer: 2
a(a)
Answer: 3
a(b)
Answer: 4
b(a)

Application: Partner Unit Problem

) y
7 £
SO Dr. G Schmitz CC BY-SA’?‘Q © Siemens

Subproblem of configuring railway safety systems

Partner Unit Problem

zone2sensor

Zzone

0.2

1.

1 .
unit

unit2zone

1.1

—X

0.2

1 *

on U
u

If zone Z is connected to sensor S then
Z is connected to a unit U and S is connected to this unit U or
Z is connected to a unit U1 and S is connected to a different unit U2
and U1 is connected to U2.

0..n

nit2unit

unit2sensor

SenNnsor

n .. fixed natural number, e.g., 2

UNIVERSITAT
KLAGENFURT

PUP in Clingo

% Facts (CUSTOMER REQUIREMENTS)

zone2sensor (zl,sl).
zone2sensor (zl,s2).
zone2sensor (zl,s3).
zone2sensor (z2,s3) .

zone2sensor (z2,s4) .

#const lower=2.
#const upper=4.
#const maxPU=2.

PUP in Clingo

% Rules (CONFIGURATION REQUIREMENTS)

comUnit (1. .upper).

zone (Z) :- zone2sensor(Z,S).
sensor (S) :- zone2sensor(Z,S).
1 { unit2zone(U,2) : comUnit(U)} 1 :- zone(2).

:— comUnit(U), 3 <= #count { Z: unit2zone(U,Z)}.

1l { unit2sensor(U,S) : comUnit(U)} 1 :- sensor(S).
:— comUnit(U), 3 <= #count { S : unit2sensor(U,S)}.

partnerunits (U,P) : - zone2sensor(Z,S), unit2zone(U,Z), unit2sensor(P,S), U!=P.
partnerunits (U,P) : - partnerunits(P,U).
:— comUnit(U), maxPU < f#ficount { U : partnerunits(U,P)}.
% OPTIMIZATION
unitUsed (U) : - unit2zone (U, Z).

unitUsed (U) : - unit2sensor(U,S).
:~ unitUsed(U). [1,U]

PUP in Clingo

% SOME TUNING

lower {unitUsed (U) : comUnit (U) }upper.
:— unitUsed(U), 1 < U, not unitUsed(U-1).

% SOLUTION SCHEMA

#show unit2zone/2.
#show unit2sensor/2.
#show partnerunits/2.
#show unitUsed/1.

PUP in Clingo

sensor 2

% SOLUTION

sensor 1

unitUsed (1) unitUsed (2) @7 sensor 3

unit2zone (2,zl) unit2zone(1l,z2)

unit2sensor(2,sl) unit2sensor(2,s4) sensor 4

unit2sensor(l,s2) unit2sensor(l,s3)

Input:

partnerunits(l,2) partnerunits(2,1) zone2sensor (z1,sl).
zone2sensor (zl,s2).
zone2sensor (zl,s3).

Optimization: 2 zone2sensor (z2,s3) .
zone2sensor (z2,s4) .

Wrap up

e Declarative problem solving is based on a declarative specification of correct solutions
e Declarative problem solving can dramatically reduce development and maintenance costs

e ASP solvers implement currently one of the most efficient and expressive knowledge
representation & reasoning frameworks for performing declarative problem solving

e ASP provides a decidable fragment of first-order logic including disjunction in the head of rules
extended by:
— Non-monotonic reasoning employing NAF, i.e., reasoning about the absence of information,
— Aggregation, an instance of second-order reasoning,
— Weak constraints, providing means for optimization

Appendix

Weak constraints — Example

employee (a) . employee (b). employee(c).
employee (d) . employee (e) .

know(a,b). know(b,c). know(c,d). know(d,e).
same skill (a,b).

married(c,d).

member (X,pl) :- employee (X), not member (X,p2).
member (X,p2) :- employee (X), not member (X,pl).
:~ member (X,P), member(Y,P), X !'= Y, not know(X,Y). [1@1,X,Y,P]
:~ member (X,P), member(Y,P), X !'= Y, married(X,Y). [102,X,Y,P]
:~ member (X,P), member(Y,P), X != Y, same skill(X,Y). [102,X,Y,P]

Weak constraints — Example

Best answer set (only member atoms):
{ member (b,pl) , member (c,pl),
member (a,p2) , member (d,p2) , member (e,p2) }

Weight=0 at level 2, weight=6 at level 1
not know(X,Y): (c¢,b), (a,d), (a,e), (d,a), (e,a), (e,d)

Sub optimal answer set (only member atoms):
{ member (b,pl) , member (c,pl), member (e,pl),
member (a,p2) , member (d,p2)}

Weight=0 at level 2, weight=7 at level 1
not know(X,Y): (b,e), (c¢,b), (c,e), (e,b), (e,c), (a,d), (d,a)

Note: there is a symmetric best model to first solution, e.g., exchange p1/p2

Semantics of logic programs with negation
Two approaches

Single intended model approach:
e Select a single model of all classical models

e Agreement for so-called “stratified programs”: Perfect model

Multiple preferred model approach:
e Select a subset of all classical models

e Different selection principles for non-stratified programs

Stratified negation

Intuition: For evaluating the body of a rule containing not r (t),
the value of the ,,negative‘ atoms r(t) should be known. Let(t) ... (t;, ... ,t,)

© Evaluate first r(t)
@ if r(t) is false, then not r{t) is true,
© if r(t) is true, then not r(t) is false and rule is not applicable

Example:
compliantBottle (axel,a),
bottleChosen (X) :- not bottleSkipped(X), compliantBottle(Y,X).

Computed model
M={ compliantBottle (axel,a), bottleChosen(a) }.

www.aau.at "l



'llUNIVERSITAT
KLAGENFURT

Program layers

e Evaluate predicates bottom up in layers
e Methods works if there is no cyclic negation (layered negation)

Example:

LO: compliantBottle (axel,a). wineBottle(a). expensive(a).

LO: bottleSkipped(X) :- expensive (X), wineBottle (X).

L1: bottleChosen(X) :- not bottleSkipped(X), compliantBottle(Y,X).

Unique (preferred) model resulting by layered evaluation (“perfect model”):

M = {compliantBottle (axel,a), wineBottle (a) ,expensive(a), bottleSkipped(a) }

Note: semantics defined by a procedure (violates declarativity)

Multiple preferred models

Unstratified Negation makes layering ambiguous:

LO0: compliantBottle (axel,a).
L?: bottleChosen(X) :- not bottleSkipped(X), compliantBottle(Y,X).
L?: bottleSkipped(X) :- not bottleChosen(X), compliantBottle(Y,X).

e Assign to a program (theory) not one but several intended models!
For instance: Answer sets!
e How to interpret these semantics? Answer set programming caters
for the following views:
O sceptical reasoning: only take entailed answers, i.e., true in all models

® brave reasoning: each model represents a different solution to the problem
© additionally: one can define to consider only a subset of preferred models

Disjunctive ASP

e The use of disjunction in rule heads is natural
man (X) | woman (X) :- person (X).
e ASP has thus been extended with disjunction
a,|a,|---|ax:-by, ..., by, notc,...,notc,.
P’

e Theinterpretation of disjunction is “minima

e Disjunctive rules thus permit to encode choices

Social dinner example - disjunctive version

Replace the choice rules

bottleSkipped(X) :- not bottleChosen(X), compliantBottle(Y,X).
bottleChosen (X) :- not bottleSkipped(X), compliantBottle(Y,X).

with an equivalent (w.r.t. example) disjunctive rule

bottleSkipped (X) | bottleChosen(X) :- compliantBottle(Y,X).

e Very often, disjunction corresponds to such cyclic negation
e However, disjunction is more expressive in general, and cannot be efficiently eliminated

Answer sets of disjunctive programs

Define answer sets similar as for normal logic programs by Gelfond-Lifschitz Reduct PM

Extend PM to disjunctive programs:

O remove each rule in Ground(P) with some literal “not a” in the body if a € M
® remove all literals “not a” from all remaining rules in Ground(P)

However, a single minimal model Im(PM) does not necessarily exist (multiple minimal models!)

Definition
M € HB(P) is an answer set of P if and only if M is a minimal (wrt. €) model of PM

Example

(1) compliantBottle (axel,a).
wineBottle (a).

(2) bottleSkipped(a) | bottleChosen(a) :- compliantBottle(axel,a).
(3) hasBottleChosen (axel) :— bottleChosen(a), compliantBottle (axel,a).

This program contains no “not”, so PM = P for every M
Its answer sets are its minimal models:

e Mi1={ wineBottle(a), compliantBottle (axel,a), bottleSkipped(a) }
e M2={ wineBottle(a), compliantBottle (axel,a), bottleChosen(a), hasBottleChosen (axel) }

This is the same as in the non-disjunctive version

Implementation of choice rules

Choice rule
{A; . 5 A} Ay - A NOtA L, ..., NOt A,

is translated into 2m+1 rules using new atoms A, A/, ... , A

A= Apiyy oo 5 Ay NOL AL, ... , NOtA,. % Condition
A :-A,notA’. A’ :-notA.. % Choice
A, :-AnotA.’. A, :-notA.. % Choice

Cardinality constraints

A (positive) cardinality constraint is of the form

low {A,; ... ; A} up % low and up are positive integers, A,, ... ,A, are classical atoms.

A cardinality constraint is satisfied in an answer set |
if the number of satisfied atoms of set {A,, ... ,A..}in | is between low and up (inclusive)

Frequently, conditional literals are employed: low {A:B,; ... ; A,:B.,} up
where B,, ... ,B,, are used to restrict the instantiations of variables occurring in A, ; ... ; A,

Example:
2{a(X) :b(X) }4.
b(r). b(s). b(t). b(u). b(v).

www.aau.at 'll



'llUNIVERSITAT
KLAGENFURT

Cardinality rules

Are employed to control the cardinality of subsets
A, - low{A,;...; A, ;notA,,,;..;not A}

Informal meaning:

o |If atleast low elements of low {A,; ...; A, ; not A,..,; ... ; not A} are true in an interpretation,
then add A, to this interpretation

e |ow is alower bound on the truth assignments in the body

Example:
a :- 1{b; c}. b.

has one answer set: {a, b}

Cardinality rules with bounds

A rule of the form

A,:- low{A,;...; A, ;notA,.,;..;not A} up.

corresponds to
A, - B, not C.
B: low{A, ...,A,, NOtA,,, ..., NOt A.}.

C:- up+1{A, ..., A, not A, ..., NOt A }.

Cardinality constraints in the head

A rule of the form
low {A;;... ;A up=-Ay, ... ,A,NOtA, ... ,NOtA..
corresponds to

B:-Amniy ooy Ap, NOt A, ... , NOt A,

{A;;... ; AL} - B.
C:-low{A; ... ; A} up.
:- B, not C.

